NASA TECHNICAL NOTE

NASA TN D-2771 c.1

NASA TN D-2771

COMPUTATION OF SUB-SATELLITE POINTS FROM ORBITAL ELEMENTS

by Richard H. Christ

John F. Kennedy Space Center

Cocoa Beach, Fla.

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION • WASHINGTON, D. C. • JUNE 1965

NASA TN D-2771

COMPUTATION OF SUB-SATELLITE POINTS

FROM ORBITAL ELEMENTS

By Richard H. Christ

John F. Kennedy Space Center
Cocoa Beach, Fla.

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

For sale by the Clearinghouse for Federal Scientific and Technical Information
Springfield, Virginia 22151 – Price $2.00

TABLE OF CONTENTS

TABLE OF CONTENTS (Cont'd)

COMPUTATION OF SUB-SATELLITE POINTS FROM ORBITAL ELEMENTS

By Richard H. Christ
John F. Kennedy Space Center

SUMMARY

This technical note presents the computer program used by the Computation Branch at the John F. Kennedy Space Center (KSC), NASA for computing predicted sub-satellite points of radiating satellites for launch interference purposes.

INTRODUCTION

Frequently, it is necessary to predict launch interference from radiating satellites. The predicted launch interference information is desired in the form of sub-satellite points and look angles. This technical note describes the method used by the Computation Branch of KSC for computing the sub-satellite points from Prediction Space Elements provided by the Goddard Space Flight Center (GSFC), NASA.

The orbital elements at the time of interest, t, are interpolated from the given ephemeris; the orbital elements at t are transformed to a position vector at t; and the sub-satellite points at t are computed. The output of the computer program may be then used for standard look angle computations.

Thanks are due to Mr. T. P. Gorman, Chief of the Advanced Orbital Programming Branch, GSFC for providing information on the prediction elements; and to Mr. W. N. Weston of GSFC who provided check data.

SYMBOLS

a	semimajor axis of orbit
e	eccentricity
i	inclination
Ω	right ascension of ascending node
ω	argument of perigee
M	mean anomaly
E	eccentric anomaly
P	period (anomalistic)
$\dot{P}$	period derivative
$\underline{r}$	radius vector
x, y, z	geocentric, inertial, right-handed, orthogonal position components; z is coincident with the polar axis and x is directed toward the vernal equinox
$\underline{i}, \underline{j}, \underline{k}$	unit vectors lying along the x,y,z coordinate axes, respectively
$\hat{e}$	eccentricity of reference spheroid

Φ'	geocentric latitude
Φ	geodetic latitude
λ	longitude
h	height above spheroid
$\hat{\omega}$	rotational rate of earth
EL	general term referring to orbital elements
t	time of interest
t_1, t_2, t_3	times of epochs 1, 2, and 3 respectively
R.A.	right ascension
S.T.	sidereal time
t_R	reference time
U.T.	universal time
C.U.L.	canonical unit of length = 6378.165 km
J.D.S.	Julian Date for Space

J.D.S. $\triangleq$ J.D. - 2,436,099.5 days
All times are expressed in terms of J.D.S. unless specified otherwise

COMPUTER PROGRAM EQUATIONS

The following are programmed digital computer equations:

Subprograms ELIN and EVERET

Interpolation for a(t), e(t), i(t), Ω (t), and ω (t) -

$$EL_1(t) = a(t),\ EL_{11} = a_1,\ EL_{12} = a_2,\ EL_{13} = a_3$$

$$EL_2(t) = e(t),\ EL_{21} = e_1,\ EL_{22} = e_2,\ EL_{23} = e_3$$

$$EL_3(t) = i(t),\ EL_{31} = i_1\ \ EL_{32} = i_2,\ EL_{33} = i_3$$

$$EL_4(t) = \Omega(t),\ EL_{41} = \Omega_1,\ EL_{42} = \Omega_2,\ EL_{43} = \Omega_3$$

$$EL_5(t) = \omega(t),\ EL_{51} = \omega_1,\ EL_{52} = \omega_2,\ EL_{53} = \omega_3$$

$$W = t_2 - t_1$$

$$S = \frac{t - t_2}{W}$$

$$R = 1 - S$$

$$\Delta_0^2 = EL_{i3} - 2EL_{i2} + EL_{i1}$$

$$EL_i(t) = R\,EL_{i2} + \frac{R(R^2-1)}{3!}\Delta_0^2 + S\,EL_{i1} + \frac{S(S^2-1)}{3!}\Delta_0^2$$

where $i = 1, 2, \cdots, 6$.

Subprogram AMIN

Computation of M(t) -

$$C_1 = 518{,}400$$

$$C_2 = 373.248$$

For the interval between t_i and t_{i+1}

$$M(t) = M_i + \frac{C_1}{P_i}(t - t_i) - C_2\frac{\dot{P}_i}{P_i^2}(t - t_i)^2$$

Subprogram EKEP

Solution of Kepler's equation -

$$Z = \frac{e(t)\sin M(t)}{\sqrt{e^2(t) + 1 - 2e(t)\cos M(t)}}$$

$$E(t)_1 = M(t) + Z - \frac{1}{6}Z^4\cot M(t)$$

$$E(t)_{i+1} = E(t)_i + \frac{M(t) + e(t)\sin E(t)_i - E(t)_i}{1 - e(t)\cos E(t)_i}$$

Continue iterating until $\left| E(t)_{i+1} - E(t)_i \right| < 0.2 \times 10^{-7}$ rad.

Subprogram PVOE

Components of position vector -

$$x(t) = a(t) \Big\{ [\cos E(t) - e(t)] [\cos \omega(t) \cos \Omega(t) - \sin \omega(t) \sin \Omega(t) \cos i(t)] + [1 - e^2(t)]^{\frac{1}{2}} \sin E(t) [- \sin \omega(t) \cos \Omega(t) - \cos \omega(t) \sin \Omega(t) \cos i(t)] \Big\}$$

$$y(t) = a(t) \Big\{ [\cos E(t) - e(t)] [\cos \omega(t) \sin \Omega(t) + \sin \omega(t) \cos \Omega(t) \cos i(t)] + [1 - e^2(t)]^{\frac{1}{2}} \sin E(t) [- \sin \omega(t) \sin \Omega(t) + \cos \omega(t) \cos \Omega(t) \cos i(t)] \Big\}$$

$$z(t) = a(t) \Big\{ [\cos E(t) - e(t)] [\sin \omega(t) \sin i(t)] + [1 - e^2(t)]^{\frac{1}{2}} \sin E(t) [\cos \omega(t) \sin i(t)] \Big\}$$

Subprogram GEODT

Sub-satellite points -

$$\text{R. A.}(t) = \tan^{-1} \frac{y(t)}{x(t)}$$

$$\Delta t_R = t - t_R$$

$$\lambda_0 = \text{S. T. at } 0^h \text{ U. T.}$$

$$\lambda(t) = \text{R. A.}(t) - \lambda_0 - \hat{\omega} \Delta t_R$$

$$r(t) = \sqrt{x^2(t) + y^2(t) + z^2(t)}$$

$$\Phi'(t) = \tan^{-1} \frac{z(t)}{\sqrt{x^2(t) + y^2(t)}}$$

$$a_2(t) = \frac{1}{1024\, r(t)}\left[512\, \hat{e}^2 + 128\, \hat{e}^4 + 60\, \hat{e}^6 + 35\, \hat{e}^8\right]$$

$$+ \frac{1}{32 r^2(t)}\left[\hat{e}^6 + \hat{e}^8\right] - \frac{3}{256 r^3(t)}\left[4\, \hat{e}^6 + 3\, \hat{e}^8\right]$$

$$a_4(t) = \frac{-1}{1024 r(t)}\left[64\, \hat{e}^4 + 48\, \hat{e}^6 + 35\, \hat{e}^8\right]$$

$$+ \frac{1}{16 r^2(t)}\left[4\, \hat{e}^4 + 2\, \hat{e}^6 + \hat{e}^8\right]$$

$$+ \frac{15 \hat{e}^8}{256 r^3(t)} - \frac{\hat{e}^8}{16 r^4(t)}$$

$$\Phi(t) = \Phi'(t) + a_2(t) \sin 2\Phi'(t) + a_4(t) \sin 4\Phi'(t)$$

$$h(t) = r(t) \frac{\sin \Phi'(t)}{\sin \Phi(t)} - \frac{1 - \hat{e}^2}{[1 - \hat{e}^2 \sin^2 \Phi(t)]^{\frac{1}{2}}}$$

MATHEMATICAL FORMULATION

Interpolation for a (t), e (t), i (t), Ω (t), and ω (t)

Consider the following difference table –

T	X	ΔX	$\Delta^2 X$	$\Delta^3 X$	$\Delta^4 X$
T_{-1}	X_{-1}		Δ^2_{-1}		Δ^4_{-1}
		$\Delta_{-\frac{1}{2}}$		$\Delta^3_{-\frac{1}{2}}$	
T_0	X_0		Δ^2_0		Δ^4_0
		$\Delta_{+\frac{1}{2}}$		$\Delta^3_{+\frac{1}{2}}$	
T_1	X_1		Δ^2_1		Δ^4_1
		$\Delta_{+1\frac{1}{2}}$		$\Delta^3_{+1\frac{1}{2}}$	
T_2	X_2		Δ^2_2		Δ^4_2

where $$\Delta_{-\frac{1}{2}} \triangleq X_0 - X_{-1}$$

$$\Delta^2_0 \triangleq \Delta_{+\frac{1}{2}} - \Delta_{-\frac{1}{2}}$$

Everett's Second Central Difference formula is given as

$$X_i = RX_0 + \frac{R(R^2 - 1^2)}{3!} \Delta^2_0 + \frac{R(R^2 - 1^2)(R^2 - 2^2)}{5!} \Delta^4_0$$

$$+ SX_1 + \frac{S(S^2 - 1^2)}{3!} \Delta^2_1 + \frac{S(S^2 - 1^2)(S^2 - 2^2)}{5!} \Delta^4_1 \quad (1)$$

where $S \triangleq \dfrac{T_i - T_0}{W}$, $R = 1 - S$,

and W is the time interval between entries in the table.

For our application, only three X's are available; therefore, the terms containing the fourth differences may be eliminated. Letting $\Delta^2_0 = \Delta^2_1$, and using the notation introduced in the previous section (Computer Program Equations), we have

$$EL(t) = R\,EL_2 + \frac{R(R^2 - 1)}{3!} \Delta^2_0 + S\,EL_1 + \frac{S(S^2 - 1)}{3!} \Delta^2_0 \quad (2)$$

where $$\Delta^2_0 \triangleq \Delta_{+\frac{1}{2}} - \Delta_{-\frac{1}{2}}$$

$$= X_1 - 2X_0 + X_{-1}$$

$$\Delta^2_0 = EL_3 - 2\,EL_2 + EL_1$$

This scheme was compared with a second degree least squares fit and agreement to 0.5×10^{-4} degree was reached for $\Omega(t)$.

Computation of M (t)

The mean anomaly at t, M (t), assuming a constant period, P, is determined from Kepler's equation

$$M(t) = M_i + \frac{2\pi}{P}(t - t_i) \tag{3}$$

Since $\dot{P}$ is available as input, equation (3) may be expanded by Taylor's series

$$f(x) = f(a) + f'(a)(x - a) + \frac{1}{2!} f''(a)(x - a)^2 + \cdot \cdot + \frac{1}{n!} f^{(n)}(a)(x - a)^n \tag{4}$$

$$f(x) = M(t)$$

$$f(a) = M_i + \frac{2\pi}{P_i}(t - t_i) = M_i + \frac{C_1}{P_i}(t - t_i) \tag{5}$$

If $(t - t_i)$ is in days, and P_i in minutes;

$$C_1 = 57.2957795 \times 1440 \times 2\pi = 518,400$$

$$f'(a) = \frac{2\pi}{P_i} - \frac{2\pi \dot{P}_i}{P_i^2}(t - t_i)$$

$$f'(a)(x - a) = \frac{2\pi}{P_i}(t - t_i) - \frac{2\pi \dot{P}_i}{P_i^2}(t - t_i)^2 = \frac{C_1}{P_i}(t - t_i) - \frac{2\,C_2}{P_i^2}\dot{P}_i(t - t_i)^2 \tag{6}$$

If $(t - t_i)$ is in days, P_i in minutes, $\dot{P}_i$ in microdays per day;

$$C_2 = \pi \times 57.2957795 \times (1440)^2 \times 10^{-6} = 373.248$$

Substitution of equations (5) and (6) into equation (4) yields the desired expression

$$M(t) = M_i + \frac{C_1}{P_i}(t - t_i) - C_2 \frac{\dot{P}_i}{P_i^2}(t - t_i)^2 \qquad (7)$$

Equation (7) is the expression given in reference 1.

Solution of Kepler's Equation

The number of methods devised for the solution of Kepler's equation exceeds one hundred. The method presented below is an iterative scheme well suited for a high-speed digital computer.

$$M(t) = E(t) - e(t) \sin E(t) \qquad (8)$$

It is desired to solve equation (8) for E (t).

The Newton-Raphson formula is given as

$$x_{i+1} = x_i - \frac{f(x_i)}{f'(x_i)} \qquad (9)$$

Application of equation (9) to equation (8) gives

$$E(t)_{i+1} = E(t)_i + \frac{M(t) + e(t) \sin E(t)_i - E(t)_i}{1 - e(t) \cos E(t)_i} \qquad (10)$$

Encke's first approximation is very good, seen as

$$E(t)_1 = M(t) Z - \frac{1}{6} Z^4 \cot M(t)$$

where

$$Z = \frac{e(t) \sin M(t)}{\sqrt{e^2(t) + 1 - 2\, e(t) \cos M(t)}}$$

Continue iterating equation (10) until

$$\left| E(t)_{i+1} - E(t)_i \right| < 0.2 \times 10^{-7} \text{ rad.}$$

Position Vector at t

The following equations (11 - 15) are taken from reference 1.

$$\underline{\Omega}(t) = \cos\Omega(t)\,\underline{i} + \sin\Omega(t)\,\underline{j} \tag{11}$$

$$\underline{\alpha}(t) = \cos i(t)\,\underline{k} + [\sin i(t)][\underline{\Omega}(t) \times \underline{k}] \tag{12}$$

$$\underline{p}(t) = \cos\omega(t)\,\underline{\Omega}(t) + [\sin\omega(t)][\underline{\alpha}(t) \times \underline{\Omega}(t)] \tag{13}$$

$$\underline{q}(t) = \underline{\alpha}(t) \times \underline{p}(t) \tag{14}$$

$$\underline{r}(t) = a(t)\left\{[\cos E(t) - e(t)]\,\underline{p}(t) + \sqrt{1 - e^2(t)}\,\sin E(t)\,\underline{q}(t)\right\} \tag{15}$$

Performing the indicated operations in equations (11–14) and substituting in equation (15) gives the desired expression for $\underline{r}(t)$ as

$$\begin{aligned} \underline{r}(t) = \underline{i}\; a(t)\Big\{ & [\cos E(t) - e(t)]\,[\cos\omega(t)\cos\Omega(t) \\ & - \sin\omega(t)\sin\Omega(t)\cos i(t)] \\ & + [1 - e^2(t)]^{\frac{1}{2}} \sin E(t)\,[-\sin\omega(t)\cos\Omega(t) \\ & - \cos\omega(t)\sin\Omega(t)\cos i(t)]\Big\} \\ + \underline{j}\; a(t)\Big\{ & [\cos E(t) - e(t)]\,[\cos\omega(t)\sin\Omega(t) \\ & + \sin\omega(t)\cos\Omega(t)\cos i(t)] \\ & + [1 - e^2(t)]^{\frac{1}{2}} \sin E(t)\,[-\sin\omega(t)\sin\Omega(t) \\ & + \cos\omega(t)\cos\Omega(t)\cos i(t)]\Big\} \\ + \underline{k}\; a(t)\Big\{ & [\cos E(t) - e(t)]\,[\sin\omega(t)\sin i(t)] \\ & + [1 - e^2(t)]^{\frac{1}{2}} \sin E(t)\,[\cos\omega(t)\sin i(t)]\Big\} \end{aligned} \tag{16}$$

Sub-satellite Points

The method used to compute geocentric latitude, Φ', and longitude, λ, is similar to that described in reference 2, as

$$\Phi'(t) = \tan^{-1} \frac{z(t)}{\sqrt{x^2(t) + y^2(t)}} \tag{17}$$

$$R.A.(t) = \tan^{-1} \frac{y(t)}{x(t)} \tag{18}$$

$$\lambda_o = S.T. \text{ at } 0^h \text{ U.T.} \tag{19}$$

then

$$\lambda(t) = R.A.(t) - \lambda_o - \hat{\omega}\,\Delta t_R$$

where

$$\Delta t_R = t - t_R \tag{20}$$

For the determination of geodetic latitude, Φ, the first two terms of the non-iterative scheme presented in reference 3 are used as

$$\Phi(t) = \Phi'(t) + a_2(t) \sin 2\Phi'(t) + a_4(t) \sin 4\Phi'(t) \tag{21}$$

where

$$a_2(t) = \frac{1}{1024r(t)}\left[512\hat{e}^2 + 128\hat{e}^4 + 60\hat{e}^6 + 35\hat{e}^8\right]$$

$$+ \frac{1}{32r^2(t)}\left[\hat{e}^6 + \hat{e}^8\right] - \frac{3}{256r^3(t)}\left[4\hat{e}^6 + 3\hat{e}^8\right],$$

$$a_4(t) = \frac{-1}{1024r(t)}\left[64\hat{e}^4 + 48\hat{e}^6 + 35\hat{e}^8\right]$$

$$+ \frac{1}{16r^2(t)}\left[4\hat{e}^4 + 2\hat{e}^6 + \hat{e}^8\right]$$

$$+ \frac{15\hat{e}^8}{256r^3(t)} - \frac{\hat{e}^8}{16r^4(t)},$$

and

$$r(t) = \sqrt{x^2(t) + y^2(t) + z^2(t)}\text{ , in C. U. L.}$$

Finally for height above spheroid, h

$$h(t) = r(t)\frac{\sin\Phi'(t)}{\sin\Phi(t)} - \frac{1 - \hat{e}^2}{[1 - \hat{e}^2 \sin^2\Phi(t)]^{\frac{1}{2}}} \qquad (22)$$

COMPUTER PROGRAM OPERATING INSTRUCTIONS

Program Description

Program Identification - SPOE 1
Computer - GE 235
Program Library - 111141
Type of Coding - FORTRAN II

Program Input (From Cards)

Card No.	Columns	Information	Mode	Remarks
1	34 - 45	Date	4A3	Information on cards; 1 and 2 for identification only
2	27 - 50	Satellite ID	8A3	
3	1 - 3	t_R (yr.)	I3	Reference Time
	4 - 6	t_R (mo.)	I3	
	7 - 9	t_R (dy.)	I3	
	10 - 12	t_R (hr.)	I3	
	13 - 15	t_R (min.)	I3	
	16 - 22	t_R (sec.)	F7.3	
	23 - 25	S. T. (hr.)	I3	S. T. at 0^h U. T. (Reference U. T.)
	26 - 28	S. T. (min.)	I3	
	29 - 35	S. T. (sec.)	F7.3	
	36 - 46	t_R	F11.5	Reference time in J. D. S.
	47 - 61	Δt	E15.9	Time increment desired in days

Program Input (From Cards) (Cont'd)

Card No.	Columns	Information	Mode	Remarks
4	1 - 15	$\hat{\omega}$	E15.9	Rotational rate of earth in radians per second
	16 - 30	$\hat{e}^2$	E15.9	$(\text{Eccentricity})^2$ of spheroid
5	1 - 11	t_1	F11.5	Epoch times in J. D. S.
	12 - 22	t_2	F11.5	
	23 - 33	t_3	F11.5	
	34 - 44	P_1	F11.5	Period in minutes
	45 - 55	P_2	F11.5	
	56 - 66	P_3	F11.5	
6	1 - 11	$\dot{P}_1$	F11.5	Period derivative in microdays per day
	12 - 22	$\dot{P}_2$	F11.5	
	23 - 33	$\dot{P}_3$	F11.5	
7	1 - 11	a_1	F11.8	Semimajor axis of orbit in earth radii
	12 - 22	a_2	F11.8	
	23 - 33	a_3	F11.8	
	34 - 44	e_1	F11.8	Eccentricity of orbit
	45 - 55	e_2	F11.8	
	56 - 66	e_3	F11.8	
8	1 - 11	i_1	F11.6	Inclination in degrees
	12 - 22	i_2	F11.6	
	23 - 33	i_3	F11.6	
	34 - 44	Ω_1	F11.6	Right ascension of ascending node in degrees
	45 - 55	Ω_2	F11.6	
	56 - 66	Ω_3	F11.6	
9	1 - 11	ω_1	F11.6	Argument of perigee in degrees
	12 - 22	ω_2	F11.6	
	23 - 33	ω_3	F11.6	

Program Input (From Cards) (Cont'd)

Card No.	Columns	Information	Mode	Remarks
	34 - 44	M_1	F11.6	Mean anomaly in degrees
	45 - 55	M_2	F11.6	
	56 - 66	M_3	F11.6	
10	1 - 3	CNTRL	F3.0	Output control card -1. print only 0. write tape only +1. print and write tape

Program Output

Option 1: Print only.
Option 2: Write tape only; output tape on plug 2, unit 1.
Option 3: Print and write tape; output tape on plug 2, unit 1.

Card 10 specifies output option. Output record is t, Φ (t), λ (t), and h (t). Input data are printed on all options.

Sample Test Case

Input. - The following listing is a computer printout of a sample input giving the Prediction Space Elements supplied by GSFC for program input:

```
DEC 28 1964

SATELLITE          RELAY 2

64 12 01 00 00 00.000 04 39 30.641 2631.00000 .600000000E-02

.729211590E-04 .669342162E-02

2631.00000 2639.00000 2647.00000 0194.71578 0194.71538 0194.71497

-0000.03520-0000.03537-0000.03559

1.74488160 1.74487910 1.74487650 0.23957545 0.23950386 0.23943548

046.326254 046.327234 046.328169 236.668682 227.838464 219.008515

172.065590 180.909426 189.755772 318.158721 016.916503 075.718748

+1.
```

Output. - The following listing is a sample computer printout of sub-satellite points.

```
                    JOHN F. KENNEDY SPACE CENTER
                        COMPUTATION BRANCH
                       SUB-SATELLITE POINTS
                           DEC 28 1964
                     SATELLITE         RELAY 2

REFERENCE TIME                                    S.T. AT 0 HR U.T.
                    YR MO DY HR MM SS.SSS          HR MM SS.SSS
 CAL DAT   UT2W     64 12  1  0  0  0.             4 39 30.641

 J.D.S.    UT2W     2631.00000

PREDICTION SPACE ELEMENTS FROM GODDARD

EPOCH                      T-ONE              T-TWO              T-THREE

J.D.S.         UT2W      2631.00000          2639.00000          2647.00000

PERIOD          MIN       194.71578           194.71538           194.71497

PERIOD DER     MD/D       -0.03520            -0.03537            -0.03559

ECCENTRICITY               0.23957545          0.23950386          0.23943548

INCLINATION     DEG       46.326254           46.327234           46.328169

RA ASC NODE     DEG      236.668682          227.838464          219.008515

ARG PERIGEE     DEG      172.065590          180.909426          189.755772

MEAN ANOMALY    DEG      318.158721           16.916503           75.718748

SEMIMAJ AXIS     ER        1.74488160          1.74487910          1.74487650
```

TIME[J.D.S.]	LAT[DEG]	LON[DEG]	HEIGHT[KM]
2631.00000	43.78	-78.84	3148.1
2631.00600	33.71	-54.73	2532.7
2631.01199	17.52	-35.00	2155.0
2631.01799	-1.62	-18.18	2110.4
2631.02399	-20.03	-1.63	2412.3
2631.02998	-34.64	16.99	2981.5
2631.03598	-43.59	38.58	3699.8
2631.04198	-46.44	61.06	4462.9
2631.04797	-44.44	80.56	5198.0
2631.05397	-39.56	95.45	5859.1
2631.05997	-33.30	106.40	6418.9
2631.06596	-26.45	114.65	6861.8
2631.07196	-19.36	121.20	7178.5
2631.07796	-12.18	126.71	7364.3
2631.08395	-4.92	131.70	7416.7
2631.08995	2.43	136.54	7335.2
2631.09595	9.91	141.62	7120.8
2631.10194	17.52	147.36	6776.5
2631.10794	25.24	154.36	6307.9
2631.11394	32.90	163.53	5725.4
2631.11993	39.97	176.25	5047.0
2631.12593	45.18	-165.69	4303.0
2631.13193	46.08	-141.85	3544.0
2631.13792	40.08	-116.26	2848.7
2631.14392	26.91	-94.15	2325.9
2631.14992	8.92	-76.07	2090.4

TIME[J.D.S.]	LAT[DEG]	LON[DEG]	HEIGHT[KM]
2631.15591	-10.33	-59.69	2206.7
2631.16191	-27.33	-42.43	2642.9
2631.16791	-39.51	-22.39	3294.0
2631.17390	-45.62	0.10	4043.6
2631.17990	-46.00	21.65	4801.2
2631.18590	-42.43	39.07	5507.3
2631.19189	-36.77	51.99	6125.2
2631.19789	-30.16	61.56	6633.8
2631.20389	-23.17	68.93	7020.6
2631.20988	-16.03	74.93	7278.7
2631.21588	-8.80	80.15	7404.5
2631.22188	-1.51	85.02	7396.5
2631.22787	5.89	89.92	7254.8
2631.23387	13.43	95.24	6981.3
2631.23987	21.10	101.48	6580.0
2631.24586	28.82	109.33	6058.4
2631.25186	36.30	119.92	5429.8
2631.25786	42.72	134.87	4717.0
2631.26385	46.31	155.73	3958.5
2631.26985	44.28	-178.90	3216.6
2631.27585	34.85	-154.51	2584.3
2631.28185	19.09	-134.45	2178.3
2631.28784	0.06	-117.49	2098.9
2631.29384	-18.56	-101.00	2369.2
2631.29984	-33.61	-82.60	2917.3
2631.30583	-43.09	-61.17	3626.1

TIME[J.D.S.]	LAT[DEG]	LON[DEG]	HEIGHT[KM]
2631.31183	-46.41	-38.56	4388.6
2631.31783	-44.72	-18.66	5128.7
2631.32382	-39.98	-3.39	5798.4
2631.32982	-33.78	7.84	6369.0
2631.33582	-26.94	16.27	6823.8
2631.34181	-19.86	22.92	7153.2
2631.34781	-12.67	28.50	7351.9
2631.35381	-5.41	33.51	7417.4
2631.35980	1.93	38.34	7349.1
2631.36580	9.38	43.39	7147.6
2631.37180	16.98	49.06	6815.8
2631.37779	24.68	55.93	6359.0
2631.38379	32.34	64.87	5787.1
2631.38979	39.47	77.22	5116.9
2631.39578	44.88	94.76	4377.5
2631.40178	46.24	118.15	3616.9
2631.40778	40.89	143.77	2911.1
2631.41377	28.27	166.24	2366.0
2631.41977	10.58	-175.43	2098.3
2631.42577	-8.71	-159.01	2180.0
2631.43176	-26.04	-141.89	2588.8
2631.43776	-38.72	-122.07	3224.2
2631.44376	-45.35	-99.63	3968.7
2631.44975	-46.14	-77.80	4728.6
2631.45575	-42.80	-59.97	5441.5
2631.46175	-37.23	-46.70	6069.3

Main Program SPOE1





Subprogram ELIN



Subprogram EVERET



Subprogram AMIN



Subprogram EKEP



Subprogram PVOE



Subprogram GEODT




21
1. COMPUTE $\bar{\omega}\Delta t_R$, MOD 2π
2. COMPUTE λ(t)
TEST λ(t)
λ(t) < 0
27
λ(t) = 0
24
λ(t) > 0
29
27
TEST [λ(t) + π]
[λ(t) + π] = 0
24
[λ(t) + π] < 0
ADD 2π TO λ(t)
24
[λ(t) + π] > 0
24
29
TEST [λ(t) − π]
[λ(t) − π] = 0
24
[λ(t) − π] > 0
SUBTRACT 2π FROM λ(t)
24
[λ(t) − π] < 0
24
24
COMPUTE φ'(t)
COMPUTE r(t), a₂(t) & a₄(t)
COMPUTE φ(t) & h(t)
26
26
RETURN TO MAIN PROGRAM

COMPUTER PROGRAM LISTING

```
*     FORTRAN
C     SPOE1- COMPUTATION OF SUB-SATELLITE POINTS FROM ORBITAL ELEMENTS
C     PROGRAMMED BY RICHARD H. CHRIST
C     UTILIZES - FUNCTION AMIN
C                FUNCTION EKEP
C                SUBROUTINE ELIN
C                FUNCTION EVERET
C                SUBROUTINE PVOE
C                SUBROUTINE GEODT
      DIMENSION EL(6,3),AL(5,3),A1(4),A2(8)
C     READ INPUT DATA
      READ 123,A1
      READ 124,A2
      READ 100,IYR,IMO,IDY,IHR,IMM,SS,ISHR,ISMM,SSS,RT,DELTAT
      READ 101,VE,ES
      READ 102,T1,T2,T3,P1,P2,P3
      READ 102,D1,D2,D3
      READ 103,((EL(I,J),J=1,3),I=1,2)
      READ 104,((EL(I,J),J=1,3),I=3,6)
      READ 129,CNTRL
C     PRINT INPUT DATA
      PRINT 105
      PRINT 121
      PRINT 122
      PRINT 127
      PRINT 123,A1
      PRINT 124,A2
      PRINT 125
```

```
      PRINT 106
      PRINT 107
      PRINT 108,IYR,IMO,IDY,IHR,IMM,SS ,ISHR,ISMM,SSS
      PRINT 120,RT
      PRINT 109
      PRINT 110
      PRINT 111,T1,T2,T3
      PRINT 112,P1,P2,P3
      PRINT 113,D1,D2,D3
      PRINT 114,(EL(2,J),J=1,3)
      PRINT 115,(EL(3,J),J=1,3)
      PRINT 116,(EL(4,J),J=1,3)
      PRINT 117,(EL(5,J),J=1,3)
      PRINT 118,(EL(6,J),J=1,3)
      PRINT 119,(EL(1,J),J=1,3)
C     FORMAT STATEMENTS TO READ AND PRINT INPUT
  100 FORMAT (3I3,2(2I3,F7.3),F11.5,E15.9)
  101 FORMAT (2E15.9)
  102 FORMAT (6F11.5)
  103 FORMAT (6F11.8)
  104 FORMAT (6F11.6)
  105 FORMAT (1H1)
  106 FORMAT (58HREFERENCE TIME                              S.T. AT 0 HR U
     1.T.)
  107 FORMAT (16X,38H YR MO DY HR MM SS.SSS    HR MM SS.SSS)
  108 FORMAT (16H CAL DAT  UT2W  ,5I3,F7.3,3X,2I3,F7.3)
  109 FORMAT (///38HPREDICTION SPACE ELEMENTS FROM GODDARD)
  110 FORMAT (/72HEPOCH                 T-ONE               T-TWO
```

```
 1             T-THREE)
111 FORMAT (/17HJ.D.S.        UT2W,4X,3(F11.5,10X))
112 FORMAT (/17HPERIOD         MIN,4X,3(F11.5,10X))
113 FORMAT (/17HPERIOD DER   MD/D,4X,3(F11.5,10X))
114 FORMAT (/17HECCENTRICITY      ,7X,3(F11.8,10X))
115 FORMAT (/17HINCLINATION    DEG,5X,3(F11.6,10X))
116 FORMAT (/17HRA ASC NODE    DEG,5X,3(F11.6,10X))
117 FORMAT (/17HARG PERIGEE    DEG,5X,3(F11.6,10X))
118 FORMAT (/17HMEAN ANOMALY   DEG,5X,3(F11.6,10X))
119 FORMAT (/17HSEMIMAJ AXIS    ER,7X,3(F11.8,10X))
120 FORMAT (/16H J.D.S.    UT2W   ,F11.5)
121 FORMAT (24X,28HJOHN F. KENNEDY SPACE CENTER)
122 FORMAT (29X,18HCOMPUTATION BRANCH)
123 FORMAT (33X,4(A3))
124 FORMAT (26X,8(A3))
125 FORMAT (////)
127 FORMAT (28X,20HSUB-SATELLITE POINTS)
129 FORMAT (F3.0)
C     FORMAT STATEMENTS FOR OUTPUT
126 FORMAT (56HTIME(J.D.S.)       LAT(DEG)       LON(DEG)       HEIGHT(KM
 1))
128 FORMAT (/(F12.5,2(7XF7.2),6X,F10.1))
C     COMPUTE NO. OF TIMES
      POINTS=(T3-T1)/DELTAT
      IF(CNTRL) 500,510,510
510 WRITE TAPE 7, POINTS, T3, T1, DELTAT
500 CONTINUE
C     CONVERT INPUT DATA AND INITIALIZE
```

```
      RAD=57.2957795
      CUL=6378.165
      SHR=ISHR
      SMM=ISMM
      ST=(SHR+((SMM+SSS/60.)/60.))*15./RAD
      PRINT 105
      PRINT 126
      DO 200 I=1,5
      DO 200 J=1,3
  200 AL(I,J)=EL(I,J)
      T=T1
      LL=POINTS
      L=0
      DO 900 N=1,LL
C     COMPUTE ORBITAL ELEMENTS AT T
      CALL ELIN(T,T1,T2,T3,AL,SAT,SET,SIT,COT,SOT)
      AMT=AMINF(T,T1,T2,T3,EL(6,1),EL(6,2),EL(6,3),P1,P2,P3,D1,D2,D3)
C     TEST MEAN ANOMALY TO SEE IF ZERO
      IF(AMT) 5,950,5
    5 AMT=AMT/RAD
C     COMPUTE ECCENTRIC ANOMALY AT T
      ET=EKEPF(AMT,SET)
      SIT=SIT/RAD
      COT=COT/RAD
      SOT=SOT/RAD
C     COMPUTE INERTIAL COORDINATES AT T
      CALL PVOE(SIT,SAT,SET,SOT,COT,ET,X,Y,Z)
      DELTR=(T-RT)*86400.
```

```
C       COMPUTE LAT., LON. AND H
        CALL GEODT(X,Y,Z,ST,DELTR,VE,ES,ELON,RHO,GLAT1,GLAT2,H,RA,GST)
        ELON=ELON*RAD
        GLAT1=GLAT1*RAD
        GLAT2=GLAT2*RAD
        RA=RA*RAD
        GST=GST*RAD
        H=H*CUL
C       PRINT AND/OR WRITE OUTPUT
        IF(CNTRL)420,405,410
    405 WRITE TAPE 7,T,GLAT2,ELON,H
        T=T+DELTAT
        GO TO 900
    410 WRITE TAPE 7,T,GLAT2,ELON,H
    420 PRINT 128,T,GLAT2,ELON,H
        T=T+DELTAT
        L=L+2
        IF (L-50) 900,900,300
    300 PRINT 105
        PRINT 126
        L=0
    900 CONTINUE
    950 STOP
        END


*       FORTRAN
C       ELIN-SUBROUTINE TO INTERPOLATE FOR ECCEN., INCL.,LON.
```

```
C         OF ASC. NODE, ARG. OF PERIGEE, SEMI MAJOR AXIS
C         UTILIZES EVERET FUNCTION SUBPROGRAM
      SUBROUTINE ELIN(T,T1,T2,T3,AL,SAT,SET,SIT,COT,SOT)
      DIMENSION AL(5,3),ELT(5)
      DO 5 I=1,5
    5 ELT(I)=EVERET(T,T1,T2,T3,AL(I,1),AL(I,2),AL(I,3))
      SAT=ELT(1)
      SET=ELT(2)
      SIT=ELT(3)
      COT=ELT(4)
      SOT=ELT(5)
      RETURN
      END


*     FORTRAN
C     EVERET-FUNCTION SUBPROGRAM TO INTERPOLATE FOR 1 VALUE
C           GIVEN 3  USING EVERETTS 2ND CENTRAL DIFFERENCE
C           FORMULA
      FUNCTION EVERET(T,T1,T2,T3,X1,X2,X3)
      W=T2-T1
      S=(T-T2)/W
      R=1.-S
      D=X3-2.*X2+X1
      XI=R*X2+R*(R*R-1.)*D/6.+S*X3+S*(S*S-1.)*D/6.
      EVERET=XI
      RETURN
      END
```

```
*     FORTRAN
C     AMIN - FUNCTION SUBPROGRAM FOR
C              MEAN ANOMALY INTERPOLATION
C
      FUNCTION AMIN(T,T1,T2,T3,AM1,AM2,AM3,P1,P2,P3,PDOT1,PDOT2,PDOT3)
      C1=518400.
      C2=373.248
      IF (T-T2) 10,20,30
   10 GO TO 80
   20 AMT=AM2
      GO TO 90
   30 IF (T-T3) 40,50,60
   40 GO TO 81
   50 AMT=AM3
      GO TO 90
   60 AMT=0.
      GO TO 90
   80 DELT=T-T1
      AMI=AM1
      PI=P1
      PDOTI=PDOT1
      GO TO 82
   81 DELT=T-T2
      AMI=AM2
      PI=P2
      PDOTI=PDOT2
```

```
82 AMT=AMI+C1/PI*DELT-C2*PDOTI/(PI*PI)*DELT*DELT
   MODPI=AMT/360.
   FLMOD=MODPI
   AMT=AMT-FLMOD*360.
90 AMIN=AMT
C  AMIN IS ZERO WHEN T IS GREATER THAN T3
   RETURN
   END
```

```
*  FORTRAN
C  EKEP-FUNCTION SUBPROGRAM TO SOLVE KEPLERS EQ. FOR AN ELLIPSE
   FUNCTION EKEP  (AM,ECC)
   DIMENSION E(10)
   TOL=.00000002
   Z=ECC*SINF(AM)/SQRTF(ECC*ECC+1.-2.*ECC*COSF(AM))
   COTAM=COSF(AM)/SINF(AM)
   E(1)=AM+Z-Z**4*COTAM/6.
   I=1
10 E(I+1)=E(I)+(AM+ECC*SINF(E(I))-E(I))/(1.-ECC*COSF(E(I)))
   DELTAE=ABSF(E(I+1)-E(I))
   IF(DELTAE-TOL) 30,30,2C
20 I=I+1
   GO TO 10
30 EKEP= E(I+1)
   RETURN
   END
```

```
*      FORTRAN
C      PVOE-SUBROUTINE TO COMPUTE POSITION VECTOR FROM ORBITAL ELEMENTS
C
       SUBROUTINE PVOE (SIT,SAT,SET,SOT,COT,ET,X,Y,Z)
       SISIT=SINF(SIT)
       COSIT=COSF(SIT)
       SISOT=SINF(SOT)
       COSOT=COSF(SOT)
       SICOT=SINF(COT)
       COCOT=COSF(COT'
       SIET=SINF(ET)
       COET=COSF(ET)
       A=COET-SET
       B=SQRTF(1.-SET*SET)*SIET
       X=SAT*(A*(COSOT*COCOT-SISOT*SICOT*COSIT)+B*(-SISOT*COCOT-COSOT*SIC
      1OT*COSIT))
       Y=SAT*(A*(COSOT*SICOT+SISOT*COCOT*COSIT)+B*(COSOT*COCOT*COSIT-SISO
      1T*SICOT))
       Z=SAT*(A*SISOT*SISIT+B*COSOT*SISIT)
       RETURN
       END


*      FORTRAN
C
C      GEODT- SUBROUTINE TO COMPUTE LAT., LON. AND HEIGHT
C      FROM INERTIAL X,Y,Z
```

```
      SUBROUTINE GEODT (X,Y,Z,ST,DT,VE,E2,ELON,RHO,GLAT1,GLAT2,H,RA,GST)
      PI=3.14159265
      TWOPI=6.28318531
C     COMPUTE R.A. OF SATELLITE, 0-360
      RA=ARTNF(Y,X)
      IF(RA)5,10,10
    5 RA=TWOPI+RA
      GO TO 15
   10 RA=RA
   15 CONTINUE
C     COMPUTE CORRECTION FOR ROTATION OF EARTH, 0-360
      RT=VE*DT
      IRT=RT/TWOPI
      FRT=IRT
      RT=RT-FRT*TWOPI
C     COMPUTE LONGITUDE, +OR-180
      ELON=RA-ST-RT
      IF(ELON)70,40,90
   70 IF(ELON+PI)80,40,40
   80 ELON=ELON+TWOPI
      GO TO 40
   90 IF(ELON-PI)40,40,30
   30 ELON=ELON-TWOPI
   40 CONTINUE
C     COMPUTE GEOCENTRIC LAT., +OR-90
      GLAT1=ATANF(Z/SQRTF(X*X+Y*Y))
C     COMPUTE GEODETIC LAT.
      E4=E2*E2
```

```
      E6=E2*E4
      E8=E2*E6
      RHO=SQRTF(X*X+Y*Y+Z*Z)
      RHO2=RHO*RHO
      RHO3=RHO*RHO2
      RHO4=RHO*RHO3
      A2=(512.*E2+128.*E4+60.*E6+35.*E8)/(1024.*RHO)
     1+(E6+E8)/(32.*RHO2)-3.*(4.*E6+3.*E8)/(256.*RHO3)
      A4=-(64.*E4+48.*E6+35.*E8)/(1024.*RHO)+(4.*E4+2.*E6+E8)/(16.*RHO2)
     1+(15.*E8)/(256.*RHO3)-E8/(16.*RHO4)
      GLAT2=GLAT1+A2*SINF(2.*GLAT1)+A4*SINF(4.*GLAT1)
C     COMPUTE HEIGHT IN C.U.L.
      S2GLAT=SINF(GLAT2)*SINF(GLAT2)
      H=RHO*SINF(GLAT1)/SINF(GLAT2)-(1.-E2)/SQRTF(1.-E2*S2GLAT)
      RETURN
      END
```

REFERENCES

1. Siry, Joseph W: Goddard Orbit Information Systems. X-547-64-108, June 1964.
2. Christ, Richard H: Computation of Quick Look Orbital Parameters. NASA KSC SP-61-E, August 28, 1964.
3. Morrison, John and Pines, Samuel: The Reduction from Geocentric to Geodetic Coordinates. Astronomical Journal, Vol. 66, No. 1, February, 1961.

APPROVAL

NASA TN

COMPUTATION OF SUB-SATELLITE POINTS
FROM ORBITAL ELEMENTS

ORIGINATOR:

R. H. Christ
Chief, Astrodynamics Analysis Unit

APPROVALS:

B. A. Corbett, Jr.
Chief, Analysis Section

B. G Griffin
Chief, Computation Branch

Dr. R. H. Bruns
Chief, Data Acquisition and
Systems Analysis Division

DISTRIBUTION

National Aeronautics and Space Administration

Headquarters Washington 25, D.C.	(2)
Langley Research Center Langley Field Hampton, Virginia	(2)
Goddard Space Flight Center Greenbelt, Maryland	(2)
Jet Propulsion Laboratory California Institute of Technology 4800 Oak Grove Drive Pasadena 3, California	(2)
Flight Research Center Edwards, California	(2)
Lewis Research Center 21000 Brookpark Road Cleveland 35, Ohio	(2)
Ames Research Center Moffett Field, California	(2)
John F. Kennedy Space Center Cocoa Beach, Florida	
ED 33	(20)
MT 2523	(10)

NASA-Langley, 1965

2/22/85

"The aeronautical and space activities of the United States shall be conducted so as to contribute . . . to the expansion of human knowledge of phenomena in the atmosphere and space. The Administration shall provide for the widest practicable and appropriate dissemination of information concerning its activities and the results thereof."

—National Aeronautics and Space Act of 1958

NASA SCIENTIFIC AND TECHNICAL PUBLICATIONS

TECHNICAL REPORTS: Scientific and technical information considered important, complete, and a lasting contribution to existing knowledge.

TECHNICAL NOTES: Information less broad in scope but nevertheless of importance as a contribution to existing knowledge.

TECHNICAL MEMORANDUMS: Information receiving limited distribution because of preliminary data, security classification, or other reasons.

CONTRACTOR REPORTS: Technical information generated in connection with a NASA contract or grant and released under NASA auspices.

TECHNICAL TRANSLATIONS: Information published in a foreign language considered to merit NASA distribution in English.

TECHNICAL REPRINTS: Information derived from NASA activities and initially published in the form of journal articles.

SPECIAL PUBLICATIONS: Information derived from or of value to NASA activities but not necessarily reporting the results of individual NASA-programmed scientific efforts. Publications include conference proceedings, monographs, data compilations, handbooks, sourcebooks, and special bibliographies.

Details on the availability of these publications may be obtained from:

SCIENTIFIC AND TECHNICAL INFORMATION DIVISION

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

Washington, D.C. 20546